

# Alignment for multiple group second-order factor analysis

*Tihomir Asparouhov and Bengt Muthén*

Mplus Web Note No.27

Version 1

December 26, 2025

## 1 Introduction

In this note we consider alignment for second-order factor analysis. Suppose that  $Y_{ijpg}$  is the observed  $p$ -th indicator of the  $j$ -th first-order factor for individual  $i$  in group  $g$ . The  $j$ -th first-order factor for individual  $i$  in group  $g$  is denoted by  $f_{ijg}$ . We consider a model with a single second-order factor for this illustration, and the second-order factor for individual  $i$  in group  $g$  is denoted by  $f_{ig}$ . Let  $P$  denote the number of indicators per first-order factor and  $M$  denote the number of first-order factors. For simplicity here we will assume that the number of indicators for each first-order factor is the same, but clearly this does not need to be the case. Denote the number of groups by  $G$ . With this assumption the total number of indicators in this model is  $PM$ . The multiple group second-order factor analysis is given by the following equations:

$$Y_{ijpg} = \nu_{jpg} + \lambda_{jpg}f_{ijg} + \varepsilon_{ijpg}$$

$$f_{ijg} = \alpha_{jg} + \beta_{jg}f_{ig} + \xi_{ijg}$$

$$\varepsilon_{ijpg} \sim N(0, \theta_{jpg})$$

$$\xi_{ijg} \sim N(0, \psi_{jg})$$

$$f_{ig} \sim N(\mu_g, \sigma_g)$$

The number of parameters in the above model is as follows. We have  $3GPM$  parameters:  $\nu_{jpg}, \lambda_{jpg}, \theta_{jpg}$ . We have  $3GM$  parameters:  $\alpha_{jg}, \beta_{jg}$ , and  $\psi_{jg}$ . We also have  $2G$  parameters:  $\mu_g, \sigma_g$ . The total number of parameters in this model is then  $3GPM + 3GM + 2G$ . Not all of these parameters can be identified. Certain restrictions are necessary to impose on this model for it to be identified. In first-order factor analysis there are three standard ways to restrict the parameters to yield identifiable models known as the Configural, Metric, and Scalar models. In second-order factor analysis the same concepts apply, but the situation is slightly more complex and there are some additional variations.

- **Configural**

In this model the restrictions are  $\alpha_{jg} = \mu_g = 0$  and  $\psi_{jg} = \sigma_g = 1$ . Just as for the configural first-order factor analysis, factor means and variances are not identifiable parameters, but all other parameters are identified and are group specific. The model essentially amounts to estimating a second-order factor analysis in each group separately. The number of parameters in this model is  $3GPM + GM$ . In what follows we will use an example with  $P = 3$ ,  $M = 4$ , and  $G = 3$ . With these settings the configural model has 120 parameters (40 in each group). The configural model has the property that it is nested above all other multiple group second-order factor analysis models, i.e., it yields the best model fit possible with this model structure. No other specification of the model restrictions can be made to improve on the model fit. If this model does not fit the data it is necessary to change the factor structure beyond the second-order factor model. The main problem with the configural model is that it cannot provide any group comparison information for the latent variable distribution.

- **Metric**

This is typically the first most basic attempt at establishing measurement group invariance for the first and the second-order models, i.e., the parameter restrictions in this case are that the loadings  $\lambda_{jpg} = \lambda_{jp}$  and  $\beta_{jg} = \beta_j$  are group invariant. The factor means are still fixed to zero  $\alpha_{jg} = \mu_g = 0$ , but the factor variances are fixed only in the first group  $\psi_{j1} = \sigma_1 = 1$ . The number of parameters in the model is

$2GPM + PM + GM + G - 1$ . For our example with  $P = 3$ ,  $M = 4$ , and  $G = 3$ , this amounts to 98 parameters. The role of the metric invariance model is in a way intermediary between the configural and the scalar model. It is meant to separate intercepts and loading group invariance testing. It also allows us to modify the loading group-invariance assumption until an acceptable model fit is obtained before exploring the intercept group invariance. Here there are some model variations. It is possible to define a metric model only for the first-order factor or only for the second-order factor.

- **Scalar**

The scalar model inherits the following restrictions imposed by the metric model:  $\lambda_{jpg} = \lambda_{jp}$ ,  $\beta_{jg} = \beta_j$ ,  $\psi_{j1} = \sigma_1 = 1$ . In single-order factor analysis models, the scalar model holds the indicator intercepts equal across groups, fixes the factor means in the first group, but estimates factor means in all other groups. If we attempt the same strategy for the second-order factor analysis for the factor structures on both levels, we appear to have a problem. The first-order factor intercepts are fixed to 0 in the first group, and the second-order factor analysis will need to hold the first-order factor intercepts equal across groups so we can estimate the second-order factor means, which means all first-order factor means must be fixed to 0. Therefore, here we offer three variations of the scalar model, all of which are nested within the metric model as in the single-order factor analysis. All of these scalar models can be used to construct alignment models.

- **Scalar-1**

In this model the restrictions are  $\nu_{jpg} = \nu_{jp}$  (group invariant indicator means), first-order factor means are fixed  $\alpha_{jg} = 0$ , and second-order factor means are free except in the reference group:  $\mu_1 = 0$ . The number of parameters is  $GPM + 2PM + GM + 2G - 2$ . In our example this is 76 parameters. This model allows means only for the second-order factors.

- **Scalar-2**

In this model the restrictions are  $\nu_{jpg} = \nu_{jp}$  (group invariant indicator means), first-order factor means are free except for the first group  $\alpha_{j1} = 0$ , and second-order factor means are fixed  $\mu_g = 0$ .

The number of parameters is  $GPM + 2PM + 2GM + G - 1 - M$ . In our example this is 82 parameters. This model allows means only for the first-order factors.

– **Scalar-3**

In this model the restrictions are  $\nu_{jpg} = 0$  (no indicator means), first-order factor means are free but group invariant  $\alpha_{jg} = \alpha_j$ , and the second-order factor means are free except for the reference group  $\mu_1 = 0$ . The number of parameters is  $GPM + PM + GM + M + 2G - 2$ . In our example this is 68 parameters. This model allows means for both the first- and second-order factors; however, we have sacrificed the indicator means and thus have obtained an even more restrictive model that may be most vulnerable to invariance misspecifications.

Figures 1–5 show Mplus input files for the above 5 models.

Figure 1: Configural second order

```

variable:
    names = y1-y12 g;
    grouping=g(1-3);

data:file=1.dat;

model:
    [f@0 f1-f4@0]; f@1; f1-f4@1;
    [y1-y12*0];
    f1 by y1-y3*.7;
    f2 by y4-y6*.7;
    f3 by y7-y9*.7;
    f4 by y10-y12*.7;
    f by f1-f4*1;

model 1:
    [f@0 f1-f4@0]; f@1; f1-f4@1;
    [y1-y12*0];
    f1 by y1-y3*.7;
    f2 by y4-y6*.7;
    f3 by y7-y9*.7;
    f4 by y10-y12*.7;
    f by f1-f4*1;

model 2:
    [f@0 f1-f4@0]; f@1; f1-f4@1;
    [y1-y12*0];
    f1 by y1-y3*.7;
    f2 by y4-y6*.7;
    f3 by y7-y9*.7;
    f4 by y10-y12*.7;
    f by f1-f4*1;

model 3:
    [f@0 f1-f4@0]; f@1; f1-f4@1;
    [y1-y12*0];
    f1 by y1-y3*.7;
    f2 by y4-y6*.7;
    f3 by y7-y9*.7;
    f4 by y10-y12*.7;
    f by f1-f4*1;

```

Figure 2: Metric second order

```
variable:
  names = y1-y12 g;
  grouping=g(1-3);

data:file=1.dat;

model:
  [f@0 f1-f4@0]; f@1; f1-f4@1;
  [y1-y12*0];
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  f by f1-f4*1 (1-4);

model 1: [f@0 f1-f4@0]; f@1; f1-f4@1; [y1-y12*0];
model 2: [f@0 f1-f4@0]; f*1; f1-f4*1; [y1-y12*0];
model 3: [f@0 f1-f4@0]; f*1; f1-f4*1; [y1-y12*0];
```

Figure 3: Scalar-1 second order

```
variable:
  names = y1-y12 g;
  grouping=g(1-3);

data:file=1.dat;

model:
  [f@0 f1-f4@0]; f@1; f1-f4@1;
  [y1-y12*0] (m1-m12);
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  f by f1-f4*1 (1-4);

model 2: [f1-f4@0]; f*1; f1-f4*1; [f];
model 3: [f1-f4@0]; f*1; f1-f4*1; [f];
```

Figure 4: Scalar-2 second order

```
variable:
  names = y1-y12 g;
  grouping=g(1-3);

data:file=1.dat;

model:
  [f@0 f1-f4@0]; f@1; f1-f4@1;
  [y1-y12*0] (m1-m12);
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  f by f1-f4*1 (1-4);

model 2: [f1-f4*0]; f*1; f1-f4*1; [f@0];
model 3: [f1-f4*0]; f*1; f1-f4*1; [f@0];
```



Figure 5: Scalar-3 second order

```
variable:
  names = y1-y12 g;
  grouping=g(1-3);

data:file=1.dat;

model:
  [f@0 f1-f4@0]; f@1; f1-f4@1;
  [y1-y12@0];
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  f by f1-f4*1 (1-4);
  [f1-f4*0] (m1-m4);

model 1: f@1; f1-f4@1; [f@0]; [y1-y12@0]; [f1-f4*0] (m1-m4);
model 2: f*1; f1-f4*1; [f*0]; [y1-y12@0]; [f1-f4*0] (m1-m4);
model 3: f*1; f1-f4*1; [f*0]; [y1-y12@0]; [f1-f4*0] (m1-m4);
```

## 2 Alignment

The alignment method has been expanded to structural equation models: ASEM, see Asparouhov and Muthén (2023); however, that expansion does not include second-order factor analysis. Thus, it is necessary to employ the more flexible penalized structural equation model (PSEM) methodology described in Asparouhov and Muthén (2024). PSEM-based alignment is essentially equivalent to the alignment method as presented in Asparouhov & Muthén (2014), see Section 5.1 of Asparouhov and Muthén (2024) and Figures 6 and 7 in the supplemental materials. The prior/penalty specification of PSEM corresponds to the alignment approach of holding the intercepts and loading parameters approximately equal across groups. The explicit specification of Alignment is replaced by the model prior/penalty specification.

To construct a second-order double alignment model with PSEM, where the factor models on both levels are aligned, we can use any one of the scalar models and convert the concepts of “group invariant” to “approximately group invariant” and the concept of “fixed to 0” to “approximately fixed to 0.” To set up “approximately group invariant” parameters, the block of parameters is given DIFF-ALF prior. To set up “approximately fixed to 0” parameters, the parameters are given ALF priors. All such priors have a variance (inversely proportional to the weight of the penalty function). We generally use for the variance parameter the smallest value that yields the “same” (difference less than 1) log-likelihood as the null model. The null model here is the second-order configural model given in Figure 1.

We presented 3 scalar models, each of which can be used to produce 3 alignment models. Alignment-1 method based on the Scalar-1 method, Alignment-2 is based on the Scalar-2, and Alignment-3 is based on the Scalar-3. For all 3 alignment models, using a prior variance of 1 replicated the log-likelihood of the null model in our example, and smaller variances produced worse log-likelihood; thus we will use a prior variance of 1 for all priors in this example. Using a simulation study we are also able to compare the three alignment methods in terms of recovering the correct (data generating) parameters. In a limited simulation study, the Alignment-1 method appears to be the best, the Alignment-3 method is a close second (nearly identical to Alignment-1), and Alignment-2 came out the worst. Thus we shall proceed with presenting the Alignment-1 method based on the Scalar-1 method. It should be noted, however, that with a different set of parameters or because of substantive arguments, the alternate alignment methods may be considered

valuable. Alignment-1 has the interpretation that the second-order factor is primarily responsible for the largest differences across groups, then the first-order factor differences can be used to explain any remaining differences across groups, and finally the specific indicator differences are included if these cannot be explained by the factor differences. Alignment-3 has a similar interpretation, which is probably the reason they produce similar results. Alignment-2 interpretation is that the primary difference across groups is explained by the first-order factors and the second-order factor is secondary. All of these interpretations stem from the scalar model interpretation. Here we choose to use the same prior variance for all parameters. If we are to allow different prior variances across parameters, it is likely that the “importance of factors” in explaining across-group differences will change. It is likely that Alignment-2 with unequal prior variances can become essentially equivalent to Alignment-1.

Figure 6 shows the Mplus input file for the double alignment of second-order factor analysis Alignment-1. In the fourth line we give a larger number of allowed iterations for the optimization algorithm. It is typical that the alignment optimization takes many more iterations than standard structural models. Without that option the estimation will likely not converge. We use a much stricter convergence criterion, which is also typically needed. Without that setting, the algorithm may result in an incomplete convergence and show a non-positive definite information matrix. Further below, we label all structural parameters as parameters  $a_i$  in group 1,  $b_i$  in group 2, and  $c_i$  in group 3. These labels are needed to specify the penalty/prior for the parameters. The first line in MODEL PRIOR specifies that the loading parameters for the first-order and second-order factors are approximately equal across groups, as well as the first-order factor indicator intercepts. For each of these parameters, we specify DIFF-ALF prior. This means that the differences between the parameters are approximately 0, i.e., the parameters are approximately equal. The parameters themselves do not need to be zero. The last three lines in MODEL PRIOR specify that the first-order factor intercepts are approximately zero in each group with ALF priors. We allow these to deviate from zero, but only if we cannot explain group differences with the second-order factor means. The last line in that figure is needed to produce detailed analysis for each “approximately equal parameter.” Figure 7 shows an example of that output for parameter 1. This is the first factor loading for the first-order factor. The parameter is determined to be group invariant. Such output is produced for every parameter with the DIFF prior,

i.e., specified as approximately invariant. For parameters that are meant to be approximately 0, such as the first-order factor intercepts, the needed information is the standard  $Z$ -score based significance test found in the model results.

Asparouhov and Muthén (2024) Supplemental materials Figure 7 shows an alternative way to specify multiple group alignment with PSEM which is more suitable for situations where there are many more groups. Instead of using a different letter for the different groups, the parameter labels can have a second index which makes the specification with large number of groups very compact. For small number of groups, however, it appears that using a different letter "a", "b", or "c" to indicate the group, makes the model setup more easily understandable.

The outcome of this model estimation is that we obtain a model with the best possible fit (i.e., as good as the configural model), and we are able to obtain information about differences in the factors across groups without having to adopt incorrect measurement invariance assumptions. The data in our example is generated with 12 non-invariant parameters: 4 in each group, one indicator intercept, one first-order factor intercept, one first-order loading, and one second-order loading. The only two models that are not rejected by the chi-square test of fit are the configural and the alignment model. The detailed invariance results in this analysis correctly identified 11 out of the 12 parameters as non-invariant. One additional first-order loading was identified incorrectly as non-invariant, and one first-order factor intercept was not identified as statistically significant due to lack of power.

In Mplus version 9, the PSEM methodology has been generalized to all Mplus models, including mixture models, multilevel models and model with numerical integration. Therefore the double alignment method presented here can be utilized within a broader sets of models.

We also present a simulation study for the alignment method. The data analyzed above was generated in this simulation study. The input file for the simulation study is given in Figures 8 and 9. The group-specific model population statements show the 12 non-invariant parameters. The model estimation statement is the same as in Figure 6, except that here we use the correct true values as starting values to obtain proper confidence interval coverage results. The results are presented in Figure 10. Here we show only the factor distribution parameters. The parameters are recovered relatively well. The results can be improved with larger samples, more groups, and more indicators.

Figure 6: Double alignment for second-order factor analysis using PSEM

```

variable: names = y1-y12 g;
          grouping=g(1-3);
data:file=1.dat;
analysis: iter=10000; conv=0.000001;

model:
  f1 by y1-y3*1;
  f2 by y4-y6*1;
  f3 by y7-y9*1;
  f4 by y10-y12*1;
  f by f1-f4*1;
  f@1 f1-f4@1;

model 1:
  f@1 f1-f4@1; [f@0];
  f1 by y1-y3*1 (a1-a3);
  f2 by y4-y6*1 (a4-a6);
  f3 by y7-y9*1 (a7-a9);
  f4 by y10-y12*1 (a10-a12);
  f by f1-f4*1 (a13-a16);
  [y1-y12] (a17-a28);
  [f1-f4] (a29-a32);

model 2:
  f f1-f4; [f];
  f1 by y1-y3*1 (b1-b3);
  f2 by y4-y6*1 (b4-b6);
  f3 by y7-y9*1 (b7-b9);
  f4 by y10-y12*1 (b10-b12);
  f by f1-f4*1 (b13-b16);
  [y1-y12] (b17-b28);
  [f1-f4] (b29-b32);

model 3:
  f f1-f4; [f];
  f1 by y1-y3*1 (c1-c3);
  f2 by y4-y6*1 (c4-c6);
  f3 by y7-y9*1 (c7-c9);
  f4 by y10-y12*1 (c10-c12);
  f by f1-f4*1 (c13-c16);
  [y1-y12] (c17-c28);
  [f1-f4] (c29-c32);

model prior:
  D0(1,28) DIFF(a# b# c#)~ALF(0,1);
  a29-a32~ALF(0,1);
  b29-b32~ALF(0,1);
  c29-c32~ALF(0,1);

output: align;

```

Figure 7: Mplus Diff analysis for approximately equal parameters

```

DIFF ANALYSIS FOR PARAMETERS
A1      B1      C1

Chi-square value      0.120
Degrees of freedom    2
P-value              0.942

Param   Param   Value   Value   Difference  SE      P-value
B1      A1      0.718   0.712   0.006       0.020   0.768
C1      A1      0.711   0.712  -0.001       0.023   0.966
C1      B1      0.711   0.718  -0.007       0.023   0.766

Approximate Invariance Holds For:
A1      B1      C1
Average Value Across Invariant Parameters:      0.714

Invariant Values, Difference to Average and Significance
Param   Value   Difference   SE      P-value
A1      0.712   -0.002      0.020   0.933
B1      0.718    0.004      0.021   0.842
C1      0.711   -0.003      0.022   0.906

```

Conducting simulation studies with the alignment method is somewhat more complex than simulation studies with standard structural models. Model parameters are recovered only when the optimization can not find an equivalent and better aligned set of parameter. More information on this issue can be found in Asparouhov and Muthén (2023).

For completeness, we also include model setups for Alignment-2 and Alignment-3. These are given in Figures 11 and 12. In certain modeling situations, these might be more suitable within a particular context. Specifically, if non-zero means for all factors is a key modeling feature, Alignment-3 might be preferable.

Figure 8: Double alignment for second order factor analysis using PSEM simulation study

```

montecarlo:
  names = y1-y12;
  nobs = 2000 2000 2000;
  nreps = 100;
  ngroups=3;

analysis: iter=10000; conv=0.000001;

model population:
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  y1-y12*1;
  f1-f4@1;
  f by f1-f4*1;
  f@1;

model population-g1:
  [y7*0.4]; f by f1*0.5; f2 by y6*1; [f3*0.3];

model population-g2:
  [y3*0.4]; f by f2*0.5; f3 by y7*1; [f4*0.3];
  [f*0.5]; f1*1.2 f2*.8 f3*0.7 f4*0.9 f*0.6;

model population-g3:
  [y11*0.4]; f by f4*0.5; f4 by y12*1; [f2*0.3];
  [f*-0.5]; f1*0.8 f2*1.2 f3*0.9 f4*0.7 f*0.9;

model:
  f1 by y1-y3*.7;
  f2 by y4-y6*.7;
  f3 by y7-y9*.7;
  f4 by y10-y12*.7;
  y1-y12*1;
  f1-f4@1;

model g1: f@1; [f@0]; f1-f4@1;
  f1 by y1-y3*.7 (a1-a3);
  f2 by y4-y5*.7 y6*1 (a4-a6);
  f3 by y7-y9*.7 (a7-a9);
  f4 by y10-y12*.7 (a10-a12);
  f by f1*0.5 f2-f4*1 (a13-a16);
  [y1-y6*0 y7*0.4 y8-y12*0] (a17-a28);
  [f1-f2*0 f3*0.3 f4*0] (a29-a32);

```



Figure 9: Double alignment for second order factor analysis using PSEM simulation study continued

```

model g2: f*1; f1-f4*1;
    f1 by y1-y3*.7 (b1-b3);
    f2 by y4-y6*.7 (b4-b6);
    f3 by y7*1 y8-y9*.7 (b7-b9);
    f4 by y10-y12*.7 (b10-b12);
    f by f1*1 f2*0.5 f3-f4*1 (b13-b16);
    [y1-y2*0 y3*0.4 y4-y12*0] (b17-b28);
    [f1-f3*0 f4*0.3] (b29-b32);
    [f*0.5]; f1*1.2 f2*.8 f3*0.7 f4*0.9 f*0.6;

model g3: f*1; f1-f4*1;
    f1 by y1-y3*.7 (c1-c3);
    f2 by y4-y6*.7 (c4-c6);
    f3 by y7-y9*.7 (c7-c9);
    f4 by y10-y11*.7 y12*1 (c10-c12);
    f by f1-f3*1 f4*0.5 (c13-c16);
    [y1-y10*0 y11*0.4 y12*0] (c17-c28);
    [f1*0 f2*0.3 f3-f4*0] (c29-c32);
    [f*-0.5]; f1*0.8 f2*1.2 f3*0.9 f4*0.7 f*0.9;

model prior:
    D0(1,28) DIFF(a# b# c#)~ALF(0,1);
    a29-a32~ALF(0,1);
    b29-b32~ALF(0,1);
    c29-c32~ALF(0,1);

```

Figure 10: Double alignment for second order factor analysis using PSEM simulation study results

	Population	ESTIMATES Average	Std. Dev.	S. E. Average	M. S. E.	95% Cover	% Sig Coeff
Group G1							
Intercepts							
F	0.000	0.0000	0.0000	0.0000	0.0000	1.000	0.000
F1	0.000	0.0102	0.0412	0.0155	0.0018	0.990	0.010
F2	0.000	-0.0064	0.0178	0.0225	0.0004	0.990	0.010
F3	0.300	0.3150	0.0742	0.0752	0.0057	0.950	0.940
F4	0.000	-0.0193	0.0297	0.0264	0.0012	0.980	0.020
Residual Variances							
F	1.000	1.0000	0.0000	0.0000	0.0000	1.000	0.000
F1	1.000	1.0000	0.0000	0.0000	0.0000	1.000	0.000
F2	1.000	1.0000	0.0000	0.0000	0.0000	1.000	0.000
F3	1.000	1.0000	0.0000	0.0000	0.0000	1.000	0.000
F4	1.000	1.0000	0.0000	0.0000	0.0000	1.000	0.000
Group G2							
Intercepts							
F	0.500	0.5486	0.0702	0.0730	0.0072	0.960	0.990
F1	0.000	-0.0010	0.0188	0.0193	0.0004	1.000	0.000
F2	0.000	-0.0149	0.0241	0.0237	0.0008	0.990	0.010
F3	0.000	-0.0159	0.0352	0.0283	0.0015	1.000	0.000
F4	0.300	0.2313	0.0838	0.0908	0.0117	0.930	0.760
Residual Variances							
F	0.600	0.6523	0.0697	0.0880	0.0075	0.990	1.000
F1	1.200	1.1857	0.1221	0.1219	0.0150	0.930	1.000
F2	0.800	0.7920	0.0961	0.1015	0.0092	0.970	1.000
F3	0.700	0.7798	0.1334	0.1499	0.0240	0.980	1.000
F4	0.900	0.9080	0.1123	0.1254	0.0126	0.980	1.000
Group G3							
Intercepts							
F	-0.500	-0.4971	0.0596	0.0642	0.0035	0.990	1.000
F1	0.000	-0.0037	0.0120	0.0154	0.0002	1.000	0.000
F2	0.300	0.2716	0.0641	0.0731	0.0049	0.980	0.940
F3	0.000	-0.0100	0.0225	0.0199	0.0006	1.000	0.000
F4	0.000	-0.0124	0.0236	0.0225	0.0007	0.990	0.010
Residual Variances							
F	0.900	0.9451	0.0887	0.1068	0.0098	1.000	1.000
F1	0.800	0.7839	0.0976	0.0922	0.0097	0.950	1.000
F2	1.200	1.1555	0.1477	0.1527	0.0236	0.930	1.000
F3	0.900	0.9348	0.1330	0.1333	0.0187	0.970	1.000
F4	0.700	0.7717	0.1207	0.1117	0.0196	0.950	1.000

Figure 11: Alignment-2 input file

```

variable: names = y1-y12 g; grouping=g(1-3);
data:file=1.dat;
analysis: iter=10000; conv=0.000001;

model:
  f1 by y1-y3*1;
  f2 by y4-y6*1;
  f3 by y7-y9*1;
  f4 by y10-y12*1;
  f by f1-f4*1;
  f@1 f1-f4@1;

model 1:
  f@1 f1-f4@1;
  f1 by y1-y3*1 (a1-a3);
  f2 by y4-y6*1 (a4-a6);
  f3 by y7-y9*1 (a7-a9);
  f4 by y10-y12*1 (a10-a12);
  f by f1-f4*1 (a13-a16);
  [y1-y12] (a17-a28);
  [f1-f4] (a29-a32);
  [f] (a33);

model 2:
  f f1-f4;
  f1 by y1-y3*1 (b1-b3);
  f2 by y4-y6*1 (b4-b6);
  f3 by y7-y9*1 (b7-b9);
  f4 by y10-y12*1 (b10-b12);
  f by f1-f4*1 (b13-b16);
  [y1-y12] (b17-b28);
  [f1-f4] (b29-b32);
  [f] (b33);

model 3:
  f f1-f4;
  f1 by y1-y3*1 (c1-c3);
  f2 by y4-y6*1 (c4-c6);
  f3 by y7-y9*1 (c7-c9);
  f4 by y10-y12*1 (c10-c12);
  f by f1-f4*1 (c13-c16);
  [y1-y12] (c17-c28);
  [f1-f4] (c29-c32);
  [f] (c33);

model prior:
  DO(#,1,32) DIFF(a# b# c#)~ALF(0,1);
  a33~ALF(0,1); b33~ALF(0,1); c33~ALF(0,1);
output: align;

```

Figure 12: Alignment-3 input file

```

variable: names = y1-y12 g; grouping=g(1-3);
data:file=1.dat;
analysis: iter=10000; conv=0.000001;

model:
  f1 by y1-y3*1;
  f2 by y4-y6*1;
  f3 by y7-y9*1;
  f4 by y10-y12*1;
  f by f1-f4*1;
  f@1 f1-f4@1;

model 1:
  f@1 f1-f4@1; [f@0];
  f1 by y1-y3*1 (a1-a3);
  f2 by y4-y6*1 (a4-a6);
  f3 by y7-y9*1 (a7-a9);
  f4 by y10-y12*1 (a10-a12);
  f by f1-f4*1 (a13-a16);
  [y1-y12] (a17-a28);
  [f1-f4] (a29-a32);

model 2:
  f f1-f4; [f];
  f1 by y1-y3*1 (b1-b3);
  f2 by y4-y6*1 (b4-b6);
  f3 by y7-y9*1 (b7-b9);
  f4 by y10-y12*1 (b10-b12);
  f by f1-f4*1 (b13-b16);
  [y1-y12] (b17-b28);
  [f1-f4] (b29-b32);

model 3:
  f f1-f4; [f];
  f1 by y1-y3*1 (c1-c3);
  f2 by y4-y6*1 (c4-c6);
  f3 by y7-y9*1 (c7-c9);
  f4 by y10-y12*1 (c10-c12);
  f by f1-f4*1 (c13-c16);
  [y1-y12] (c17-c28);
  [f1-f4] (c29-c32);

model prior:
  DO(#,1,16) DIFF(a# b# c#)~ALF(0,1);
  DO(#,29,32) DIFF(a# b# c#)~ALF(0,1);
  a17-a28~ALF(0,1);
  b17-b28~ALF(0,1);
  c17-c28~ALF(0,1);

output: align;

```

## References

- [1] Asparouhov, T., & Muthén, B. (2014) Multiple-group factor analysis alignment. *Structural Equation Modeling: A Multidisciplinary Journal*, 21, 495–508.
- [2] Asparouhov, T., & Muthén, B. (2023). Multiple group alignment for exploratory and structural equation models. *Structural Equation Modeling: A Multidisciplinary Journal*, 30, 169–191.
- [3] Asparouhov, T., & Muthén, B. (2024). Penalized Structural Equation Models. *Structural Equation Modeling: A Multidisciplinary Journal*, 31, 429–454.